

9/4


Mathématiques

9/4.1

Nom d'un jour de la semaine

Plusieurs formules permettent de connaître le nom d'un jour de la semaine. Nous utilisons ici celle de GAUSS.

Si JO est le jour, MO le mois, C les deux premiers chiffres de l'année et A les deux derniers chiffres de l'année, le numéro du jour est obtenu par :



Le programme écrit en BASIC est le suivant :

```
1000 REM =====
1010 REM              Nom d'un jour de la semaine
1020 REM =====
1030 GOSUB 2000 'Initialisation
1040 GOSUB 3000 'Calcul
1050 END

2000 REM =====
2010 REM              Initialisation
2020 REM =====
2030 FOR I=1 TO 7
2040   READ J$(I)
2050 NEXT I
2060 DATA Dimanche,Lundi,Mardi,Mercredi,Jeudi,Vendredi,Samedi
2070 RETURN

3000 REM =====
3010 REM              Saisie du jour et calcul
3020 REM =====
3030 MODE 1:PRINT" CALCUL DU NOM D'UN JOUR DE LA SEMAINE"
3040 LOCATE 1,5
3050 INPUT "Jour (Entre 1 et 31)   : ";JO
3060 INPUT "Mois (Entre 1 et 12)  : ";MO
3070 INPUT "Annee (sur 4 chiffres) : ";AN$
3080 M=MO-2:IF MO<=2 THEN M=MO+10
3090 C=VAL(LEFT$(AN$,2)):A=VAL(RIGHT$(AN$,2))
3100 IF MO<=2 THEN A=A-1
3110 X=INT(2.6*M-0.199)+JO+A+INT(A/4)+INT(C/4)-2*C
3120 Z=INT(X)-7*INT(X/7)
3130 PRINT:PRINT"Le";JO;"/";MO;"/ ";AN$;" est un ";J$(Z+1)
3140 RETURN
```

Lignes 1000 à 1050 : Programme principal.

Lignes 2000 à 2070 : Initialisation.

Lignes 3000 à 3140 : Saisie du jour et calculs.

